Mauszeiger als HandSymbol anzeigen
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

 FRAGE:  Ich verwende in einem Formular ein Objekt (Bild) und möchte beim Klicken auf dieses Bild eine Aktion ausführen lassen, zum Beispiel ein Formular öffnen. Dabei möchte ich, dass sich der normale Mauszeiger (Pfeil), sobald er auf dem Bild gesetzt wird, sich in eine Hand ändert. Leider habe ich die Änderungsmethode nirgendwo finden können. Können Sie mir helfen?
M. Verburg

 ANTWORT:  Access selbst bietet keine Möglichkeit, einen individuellen Mauszeiger zu setzen. Hier können Sie aber per API-Funktion Abhilfe schaffen:

· Kopieren Sie zunächst die folgenden Deklarationen und Prozeduren in den allgemeinen Teil Ihres Formulars:

Private Const IDC_HAND As Long = 32649
Private Const IDC_ARROW As Long = 32512

Private Const GCL_HCURSOR = -12

Private Declare Function LoadCursor _
        Lib "user32" _
        Alias "LoadCursorA" ( _
        ByVal hInstance As Long, _
        ByVal lpCursorName As Long) As Long

Private Declare Function SetClassLong _
        Lib "user32" _
        Alias "SetClassLongA" ( _
        ByVal hwnd As Long, _
        ByVal nIndex As Long, _
        ByVal dwNewLong As Long) As Long

Dim blnIsHand As Boolean

Sub RemoveHandCursor()
  Dim R As Long
  Dim hArrowCursor As Long

  On Error Resume Next
'Kein Hyperlink-Cursor gesetzt
  If Not blnIsHand Then Exit Sub 

  'Zu verwendenden Cursor laden
  hArrowCursor = LoadCursor(ByVal 0&, IDC_ARROW)
  'Standard-Cursor setzen
  R = SetClassLong(Me.hwnd, GCL_HCURSOR, _
                   hArrowCursor)
  DoEvents

  'Flag zurücksetzen
  blnIsHand = False

End Sub

Sub SetHandCursor()
  Dim hHandCursor As Long
  Dim R As Long

  On Error Resume Next
  If blnIsHand Then Exit Sub 'Ist schon gesetzt

  'Zu verwendenden Cursor laden
  hHandCursor = LoadCursor(ByVal 0&, IDC_HAND)
  'Neuen Cursor setzen
  R = SetClassLong(Me.hwnd, GCL_HCURSOR, hHandCursor)
  DoEvents

  'Flag setzen
  blnIsHand = True

End Sub
· Kopieren Sie folgende Anweisungen in die Ereignisprozedur "Bei Mausbewegung" des Bild-Steuerelementes:

SetHandCursor
· Kopieren Sie folgende Anweisungen in die Ereignisprozedur "Bei Mausbewegung" des Detail-Bereiches des Formulars:

RemoveHandCursor

Wenn Sie nun den Mauszeiger über das Bild bewegen, wird über die Ereignisprozedur "Bei Mausbewegung" die Prozedur "SetHandCursor" aufgerufen. Die Prozedur lädt per API-Funktion "LoadCursor" das "Hand"-Symbol in den Speicher und die API-Funktion "SetClassLong" ordnet das Symbol dem Formular als anzuzeigenden Cursor zu. Wird die Maus aus dem Bild bewegt, löst das die Ereignisprozedur "Bei Mausbewegung" des Detailbereichs aus. Dadurch wird die Prozedur "RemoveHandCursor" aufgerufen, die per "LoadCursor" den Standard-Mauszeiger "Pfeil" lädt. "SetClassLong" ordnet den Mauzeiger dann wieder dem Formular als anzuzeigenden Mauszeiger zu. Wenn Sie einen Formularkopf oder einen Formularfuß verwenden, muss dort in der Ereignisprozedur "Bei Mausbewegung" ebenfalls die Prozedur "RemoveHandCursor" wie oben erläutert aufgerufen werden!

